
Problem 3.18

Problem 3.18. Use a computer to reproduce Table 3.2 and the associated graphs of entropy, temperature, heat capacity, and magnetization. (The graphs in this section are actually drawn from the analytic formulas derived below, so your numerical graphs won't be quite as smooth.)

Two state paramagnet with 'n' dipoles, nUp spin up, and nDown spin down. The magnetic dipole moment is μ . The external magnetic field has strength B.

Reproduce Table 3.2. Introduced 'scaled' versions of U, M, and S to make table entries dimensionless. Use the N[] function to change the huge exact integer expressions into more-readable approximate numerical values.

```
In[1]:= Clear["Global`*"]
```

First, calculate the energy, magnetization, multiplicity, and entropy.

```
In[2]:=  $\Omega[n_, nUp_] := N\left[\frac{n!}{nUp! (n - nUp)!}\right]$ 
```

```
In[3]:=  $U[n_, nUp_] := \mu B (n - 2 nUp); Us[n_, nUp_] := U[n, nUp] / (\mu B)$ 
```

```
In[4]:=  $M[n_, nUp_] := N\left[-\frac{U[n, nUp]}{B}\right]; Ms[n_, nUp_] := M[n, nUp] / (n \mu)$ 
```

```
In[5]:=  $S[n_, nUp_] := k \text{Log}[\Omega[n, nUp]]; Ss[n_, nUp_] := S[n, nUp] / k$ 
```

The Temperature and heat capacity calculations involve derivatives. Approximate them by a centered finite-difference scheme for (nUp+1) and (nUp-1). However, that only works if nUp is in the range [1, ntotal-1]. Outside that range, simply call the derivative 0. Also, calculate the scaled versions indicated in the table.

```
In[6]:=  $Ts[ntotal_, nUp_] :=$   
 $\text{If}[0 < nUp < ntotal, \frac{Us[ntotal, nUp + 1] - Us[ntotal, nUp - 1]}{Ss[ntotal, nUp + 1] - Ss[ntotal, nUp - 1]}, 0]$ 
```

```
In[7]:=  $Cvs[ntotal_, nUp_] :=$   
 $\text{If}[0 < nUp < ntotal, \frac{1}{ntotal} * \frac{Us[ntotal, nUp + 1] - Us[ntotal, nUp - 1]}{Ts[ntotal, nUp + 1] - Ts[ntotal, nUp - 1]}, 0]$ 
```

```
In[8]:= ntotal = 100;
```

```
In[9]:= data = Table[{nUp, Us[ntotal, nUp], Ms[ntotal, nUp],  $\Omega$ [ntotal, nUp],
  Ss[ntotal, nUp], Ts[ntotal, nUp], Cvs[ntotal, nUp]}, {nUp, ntotal, 0, -1}];
```

Power: Infinite expression $\frac{1}{0.}$ encountered. ⓘ

Power: Infinite expression $\frac{1}{0.}$ encountered. ⓘ

Power: Infinite expression $\frac{1}{0.}$ encountered. ⓘ

General: Further output of Power::infy will be suppressed during this calculation. ⓘ

```
In[10]:= (* Show a few sample segments of the full table. *)
n2 = IntegerPart[ntotal/2];
sample = Flatten[{Range[1, 4], Range[n2 - 1, n2 + 3], Range[-4, -1]}]
```

```
Out[11]= {1, 2, 3, 4, 49, 50, 51, 52, 53, -4, -3, -2, -1}
```


```
In[12]:= TableForm[data[[sample]],
  TableHeadings → {None, {"nUp", "U/ $\mu$ B", "M/N $\mu$ ", " $\Omega$ ", "S/k", "kT/ $\mu$ B", "CV/Nk"}}]
```

```
Out[12]//TableForm=
```

nUp	U/ μ B	M/N μ	Ω	S/k	kT/ μ B	CV/Nk
100	-100	1.	1.	0.	0	0
99	-98	0.98	100.	4.60517	0.470193	0.0738833
98	-96	0.96	4950.	8.50714	0.541394	0.309889
97	-94	0.94	161700.	11.9935	0.599271	0.364569
52	-4	0.04	9.32066×10^{28}	66.7046	25.2343	0.00118777
51	-2	0.02	9.89131×10^{28}	66.764	50.4884	0
50	0	0.	1.00891×10^{29}	66.7838	ComplexInfinity	-0.00039613
49	2	-0.02	9.89131×10^{28}	66.764	-50.4884	0
48	4	-0.04	9.32066×10^{28}	66.7046	-25.2343	0.00118777
3	94	-0.94	161700.	11.9935	-0.599271	0.364569
2	96	-0.96	4950.	8.50714	-0.541394	0.309889
1	98	-0.98	100.	4.60517	-0.470193	0.0738833
0	100	-1.	1.	0.	0	0

```
In[13]:= Splot = ListLinePlot[data[[All, {2, 5}]], LabelStyle → Larger,
    AxesLabel → {"U/μB", "S/k"}, ImageSize → Scaled[0.8], PlotRange → All]
```

Out[13]=



```
In[14]:= SetDirectory[NotebookDirectory[]]
Export["Paramagnet-Splot.pdf", Splot]
```

Out[14]=


/Users/doughera/335/2026/lectures-dev/Ch03

Out[15]=

Paramagnet-Splot.pdf


```
In[22]:= ListLinePlot[Select[data[[All, {2, 6}]], #[[1]] < 0 &], LabelStyle → Larger,
  AxesLabel → {"U/μB", "kT/μB"}, ImageSize → Scaled[0.8], PlotRange → All]
```

Out[22]=




```
In[16]:= ListLinePlot[data[[All, {2, 6}]], LabelStyle → Larger,
  AxesLabel → {"U/μB", "kT/μB"}, ImageSize → Scaled[0.8], PlotRange → All]
```

Out[16]=




```
In[17]:= ListLinePlot[data[[All, {6, 7}]],
  LabelStyle → Larger, AxesLabel → {"kT/μB", "C/Nk"},
  ImageSize → Scaled[0.8], PlotRange → {{0, 10}, All}]
```

Out[17]=



```
In[18]:= ListLinePlot[data[[All, {6, 3}]],
  LabelStyle → Larger, AxesLabel → {"kT/μB", "M/Nμ"},
  ImageSize → Scaled[0.8], PlotRange → {{-10, 10}, All}]
```

Out[18]=



```
In[19]:= ListLinePlot[Map[{1/#[[6]],#[[3]]}&,data],LabelStyle->Larger,
  AxesLabel->{"μB/kT","M/Nμ"},ImageSize->Scaled[0.8],PlotRange->{All}]
```

Power: Infinite expression $\frac{1}{0}$ encountered. ⓘ

Power: Infinite expression $\frac{1}{0}$ encountered. ⓘ

ListLinePlot: The value of the option PlotRange -> {All} is not All, Full, Automatic, a positive machine number or an appropriate list of range specifications .

Out[19]=

